
50 CLS
100 '=======================================================================
200 'COMPONENT RELIABILITY DATA
300 '=======================================================================
350 Print TAB(71) "DD-MM-YYYY"
351 Print TAB(71) Date$
352 Print TAB(73) "HH:MM:SS"
353 Print TAB(73) TIME$
400 Print "  COMPONENT RELIABILITY DATA (COMPDATA.BAS)"
401 Print "  Written by Phil Rutherford circa 1985"
402 Print "  Revised by Phil Rutherford 2020-10-22"
403 Print "  Run with MMBasic.exe (www.mmbasic.com)"
500 Print
510 Print "  Input basic operability properties of a"
520 Print "  component to compute its reliability and"
530 Print "  availabilty"
1100 Print
1250 Input "  COMPONENT NAME      ", COMP$
1325 Input "  COMPONENT ACRONYM   ", ACRON$
1350 Print
1400 Print "  1. OPERATING?"
1500 Print "  2. OR STANDBY?          OS = ";
1600 Input "", OS
1650 Print
1700 Print "  1. CONTINUOUSLY MONITORED?"
1710 If OS = 1 Then MT = 1: Print "  2. PERIODICALLY TESTED?  MT = 1": Goto 1950
1800 Print "  2. PERIODICALLY TESTED?     "
1820 Print "  3. OR UNTESTED?         MT = ";
1900 Input "", MT
1950 Print
2000 Print "  1. REPAIRABLE?"
2100 Print "  2. OR NON-REPAIRABLE?   R = ";
2200 INPUT "", R
2300 IF OS = 1 AND MT = 1 AND R = 1 THEN GOTO 4000
2400 IF OS = 1 AND MT = 1 AND R = 2 THEN GOTO 9000
2700 IF OS = 2 AND MT = 1 AND R = 1 THEN GOTO 24000
2750 IF OS = 2 AND MT = 1 AND R = 2 THEN GOTO 29000
2800 IF OS = 2 AND MT = 2 AND R = 1 THEN GOTO 34000
2900 IF OS = 2 AND MT = 2 AND R = 2 THEN GOTO 39000
2920 IF OS = 2 AND MT = 3 AND R = 1 THEN GOTO 39000
2940 IF OS = 2 AND MT = 3 AND R = 2 THEN GOTO 39000
3000 GOTO 1100
4000 '---------------------------------
4010 ' OPERATING, MONITORED, REPAIRABLE
4015 '---------------------------------
4016 Print
4020 Input "  OPERATING FAILURE RATE (/HR)     ", FRH
4030 Input "  DEMAND FAILURE RATE    (/D )     ", FRD
4040 Input "  REPAIR TIME            (HR )     ", TREP
4050 Input "  GRACE PERIOD           (HR )     ", TGRA
4060 Print TAB(3) "  Press any key to continue"

4070 CMD$ = INKEY$
4080 IF CMD$ = "" THEN GOTO 4070
4083 IF FRD <> 0 THEN U = FRD: FRH = FRD / (TREP * (1 - FRD)): GOTO 4100
4090 U = FRH * TREP / (1 + FRH * TREP)
4095 FRD = U
4100 U = U * EXP(-TGRA / TREP)
4110 A = 1 - U
4115 RO = U
4116 TRES = U / FRH
4120 GOTO 50000
9000 '-------------------------------------
9010 ' OPERATING, MONITORED, NON-REPAIRABLE
9015 '-------------------------------------
9016 Print
9020 Input "  OPERATING FAILURE RATE (/HR)     ", FRH
9030 Input "  DEMAND FAILURE RATE    (/D )     ", FRD
9040 Input "  MISSION TIME           (HR )     ", TMIS
9060 Print "  Press any key to continue"
9070 CMD$ = INKEY$
9080 IF CMD$ = "" THEN GOTO 9070
9083 IF FRD <> 0 THEN U = FRD: FRH = -(LOG(1 - FRD)) / TMIS: GOTO 9110
9090 U = 1 - EXP(-FRH * TMIS)
9095 FRD = U
9110 A = 1 - U
9115 HF = U
9116 TRES = U / FRH
9120 GOTO 50000
24000 '-------------------------------
24010 ' STANDBY, MONITORED, REPAIRABLE
24015 '-------------------------------
24016 Print
24020 Input "  STANDBY FAILURE RATE (/HR)     ", FRH
24030 Input "  DEMAND FAILURE RATE  (/D )     ", FRD
24040 Input "  REPAIR TIME          (HR )     ", TREP
24050 Input "  GRACE PERIOD         (HR )     ", TGRA
24060 Print "  Press any key to continue"
24070 CMD$ = INKEY$
24080 IF CMD$ = "" THEN GOTO 24070
24083 IF FRD <> 0 THEN U = FRD: FRH = FRD / (TREP * (1 - FRD)): GOTO 24100
24090 U = FRH * TREP / (1 + FRH * TREP)
24095 FRD = U
24100 U = U * EXP(-TGRA / TREP)
24110 A = 1 - U
24115 RO = U
24116 TRES = U / FRH
24120 GOTO 50000
29000 '-----------------------------------
29010 ' STANDBY, MONITORED, NON-REPAIRABLE
29015 '-----------------------------------
29016 Print
29020 Input "  STANDBY FAILURE RATE (/HR)     ", FRH

29030 Input "  DEMAND FAILURE RATE  (/D )     ", FRD
29040 Input "  MISSION TIME         (HR )     ", TMIS
29060 Print "  Press any key to continue"
29070 CMD$ = INKEY$
29080 IF CMD$ = "" THEN GOTO 29070
29083 IF FRD <> 0 THEN U = FRD: FRH = -(LOG(1 - FRD)) / TMIS: GOTO 29110
29090 U = 1 - EXP(-FRH * TMIS)
29095 FRD = U
29110 A = 1 - U
29115 HF = U
29116 TRES = U / FRH
29120 GOTO 50000
34000 '----------------------------
34010 ' STANDBY, TESTED, REPAIRABLE
34015 '----------------------------
34016 Print
34020 Input "  STANDBY FAILURE RATE           (/HR)   ", FRH
34030 Input "  DEMAND FAILURE RATE            (/D )   ", FRD
34040 Input "  REPAIR TIME                    (HR )   ", TREP
34042 Input "  GRACE PERIOD                   (HR )   ", TGRA
34044 Input "  TEST INTERVAL                  (HR )   ", TINT
34045 Input "  TEST DURATION                  (HR )   ", TDUR
34046 Input "  TEST OVERRIDE PROBABILITY              ", TOP
34047 Input "  SCHEDULED MAINTENANCE INTERVAL (HR )   ", SMI
34048 Input "  SCHEDULED MAINTENANCE DURATION (HR )   ", SMD
34060 Print "  Press any key to continue"
34070 CMD$ = INKEY$
34080 IF CMD$ = "" THEN GOTO 34070
34083 IF FRD <> 0 THEN HF = FRD: FRH = (FRD - (SMD / SMI) - (TDUR / TINT) * (1 -
TOP)) / (TREP + TINT / 2): GOTO 34092
34087 G = FRH * TINT
34090 HF = 1 - (G - G * G / 2 + G * G * G / 6 - G * G * G * G / 24) / G
34092 TOUT = (TDUR / TINT) * (1 - TOP)
34094 RO = (FRH * TREP) * EXP(-TGRA / TREP)
34096 SMO = SMD / SMI
34097 U = HF + TOUT + RO + SMO
34098 IF FRD = 0 THEN FRD = U
34110 A = 1 - U
34116 TRES = U / FRH
34120 GOTO 50000
39000 '--------------------------------------------
39010 ' STANDBY, NON-REPAIRABLE, UNTESTED OR TESTED
39015 '--------------------------------------------
39016 Print
39020 Input "  STANDBY FAILURE RATE (/HR)     ", FRH
39030 Input "  DEMAND FAILURE RATE  (/D )     ", FRD
39040 Input "  MISSION TIME         (HR )     ", TMIS
39060 Print "  Press any key to continue"
39070 CMD$ = INKEY$
39080 IF CMD$ = "" THEN GOTO 39070
39083 IF FRD <> 0 THEN U = FRD: FRH = FRD / (TMIS / 2): GOTO 39110

39087 G = FRH * TMIS
39090 U = 1 - (G - G * G / 2 + G * G * G / 6 - G * G * G * G / 24) / G
39095 FRD = U
39110 A = 1 - U
39115 HF = U
39116 TRES = U / FRH
39120 Goto 50000
49999 '-----------------
50000 'DOS SCREEN OUTPUT
50050 '-----------------
50060 Print
50100 Print "  ----------------------------------------------------------------"
50300 Print "  COMPONENT NAME     "; COMP$
50400 Print "  ACRONYM            "; ACRON$
50450 Print
50500 If OS = 1 Then Print "  OPERATING...";  Else Print "  STANDBY...";
50600 If MT = 1 Then Print "CONTINUOUSLY MONITORED...";  Else If MT = 2 Then Print
"PERIODICALLY TESTED...";  Else Print "UNTESTED...";
50700 If R = 1 Then Print "REPAIRABLE" Else Print "NON-REPAIRABLE"
50750 Print
50800 Print "  FAILURE RATE...................(/HR)  FRH  "; FRH
51000 Print "  FAILURE RATE...................(/D )  FRD  "; FRD
51200 Print "  REPAIR TIME....................(HR )  TREP "; TREP
51330 Print "  GRACE PERIOD...................(HR )  TGRA "; TGRA
51400 Print "  TEST INTERVAL..................(HR )  TINT "; TINT
51600 Print "  TEST DURATION..................(HR )  TDUR "; TDUR
51720 Print "  TEST OVERRIDE PROBABILITY...........  TOP  "; TOP
51750 Print "  SCHEDULED MAINTENANCE INTERVAL (HR )  SMI  "; SMI
51770 Print "  SCHEDULED MAINTENANCE DURATION (HR )  SMD  "; SMD
51800 Print "  MISSION TIME...................(HR )  TMIS "; TMIS
51920 Print "  RESIDENCE TIME.................(HR )  TRES "; TRES
52000 Print "  UNAVAILABILITY......................  U    "; U
52200 Print "      HARDWARE FAILURE................  HF   "; HF
52400 Print "      TEST OUTAGE.....................  TOUT "; TOUT
52600 Print "      REPAIR OUTAGE...................  RO   "; RO
52800 Print "      SCHEDULED MAINTENANCE...........  SMO  "; SMO
53000 Print "  AVAILABILITY........................  A    "; A
53100 Print "  ------------------------------------------------------------------"
53400 Input "  Another ? (y/n) ", CMD$
53500 If CMD$ = "Y" Or CMD$ = "y" Then Goto 1100
55000 End


